Asymptotics for the simplest generalized Jacobi polynomials recurrence coefficients from Freud's equations: numerical explorations.

Alphonse Magnus,
Institut de Mathématique Pure et Appliquée,
Université Catholique de Louvain,
Chemin du Cyclotron,2,
B-1348 Louvain-la-Neuve
(Belgium)
(0)(10)473157 , alphonse.magnus@uclouvain.be ,
http://www.math.ucl.ac.be/membres/magnus/

The present file is http://www.math.ucl.ac.be/membres/magnus/num3/gjacobimagnus.pdf

Dedicated to L. Gatteschi on the occasion of his 70th birthday
published as
Annals of Numerical Mathematics **2** (1995) 311-325.

Abstract. Generalized Jacobi polynomials are orthogonal polynomials related to a weight function which is smooth and positive on the whole interval of orthogonality up to a finite number of points, where algebraic singularities occur. The influence of these singular points on the asymptotic behaviour of the recurrence coefficients is investigated.

AMS(MOS) subject classification. 42C05.

Key words. Orthogonal polynomials, generalized Jacobi weights, recurrence coefficients.

1.Weight singularities and recurrence coefficients.

The orthonormal polynomials $p_n(x) = \gamma_n x^n + \cdots$ related to the weight w satisfy the three-terms recurrence relation

$$a_{n+1}p_{n+1}(x) = (x - b_n)p_n(x) - a_n p_{n-1}(x), \tag{1}$$

with $a_0 p_{-1}(x) \equiv 0$.

Let $w(x) > 0$ hold almost everywhere on the support $[-1, 1]$, then one knows (since 1977) that the recurrence coefficients have limits $a_n \to 1/2$ and $b_n \to 0$ when $n \to \infty$ (see for instance the survey in [Nev2]).

Features of w can somehow be "read" in the sequence of the recurrence coefficients. This goes back to Stieltjes and is current practice in solid-state physics [LaG]. See [Ap] for the case of endpoints singularities. J.P. Gaspard once showed me a paper by C.Hodges [Ho] describing the influence of a mild *interior* singularity of the form

$$\begin{aligned} w(x) &\sim w(x_0) + A(x - x_0)^\gamma\,, \qquad x \to x_0, x > x_0,\\ &\sim w(x_0) + B(x_0 - x)^\gamma\,, \qquad x \to x_0, x < x_0, \end{aligned} \tag{2}$$

with $x_0 \in (-1,1), 0 < w(x) < \infty$ on $(-1,1)$ and $\gamma > 0$ (Van Hove singularity [Ho,Mart]) as

$$a_n - 1/2 \sim \xi n^{-\gamma-1}\cos(2n\theta_0 - \eta), \quad b_n \sim 2\xi n^{-\gamma-1}\cos((2n+1)\theta_0 - \eta), \quad n \to \infty$$

with $x_0 = \cos\theta_0$. I gave a lengthy proof of this in [Mag1], showing how ξ and η are related through Toeplitz determinants to the Szegő function of $w(cos\theta)\sin\theta$ (the analytic function $D(z)$ in $|z| < 1$, with $D(0) > 0$, without zero in $|z| < 1$, and such that the boundary values satisfy $|D(e^{i\theta})|^2 = w(\cos\theta)\sin\theta)$ by

$$\xi e^{i\eta} = \frac{\left(\dfrac{\sin\theta_0}{2}\right)^{\gamma+1}\Gamma(\gamma+1)}{2\pi w(x_0)} e^{i(\theta_0 - 2\arg(D(e^{i\theta_0}))}\left[Ae^{i\pi\gamma/2} - Be^{-i\pi\gamma/2}\right].$$

Hodges argument is based on the continued fraction expansion of

$$f(z) = \int_{-1}^{1}\frac{w(t)\,dt}{z-t} = \cfrac{\mu_0}{z - b_0 - \cfrac{a_1^2}{z - b_1 - \cdots}}$$

when $z = x + i\varepsilon$ for small $\varepsilon > 0$ and $-1 < x < 1$, but the fastest explanation is probably related to inverse scattering techniques ([VA] p.117, [NV] and references therein): let us consider the function q_n defined by the integral $q_n(z) = \int_{-1}^{1}(z-t)^{-1}p_n(t)w(t)dt$ when $z \notin [-1,1]$. Remark that $f(z) = \sqrt{\mu_0}q_0(z)$. From (1), $a_{n+1}q_{n+1}(z) = (z-b_n)q_n(z) - a_nq_{n-1}(z) - \sqrt{\mu_0}\delta_{n,0}$, (with $a_0q_{-1}(z) \equiv 0$), which can be written as

$$q_{n+1}(z) - 2zq_n(z) + q_{n-1}(z) = q_{n+1}(z) - \rho(z)^{-1}q_n(z) - \rho(z)[q_n(z) - \rho(z)^{-1}q_{n-1}(z)] = \varepsilon_n(z),\ n = 0,1,\dots$$

with $\varepsilon_n(z) = (1-2a_{n+1})q_{n+1}(z) - 2b_nq_n(z) + (1-2a_n)q_{n-1}(z) - 2\sqrt{\mu_0}\delta_{n,0}$, and where $\rho(z)$ is the determination of $(z + \sqrt{z^2-1})/2$ such that $|\rho(z)| > 1$ when $z \notin [-1,1]$. After a simple summation, $\rho(z)^{-N}(q_{N+1}(z) - \rho(z)^{-1}q_N(z)) - \rho(z)q_0(z) = \sum_0^N \rho(z)^{-n}\varepsilon_n(z)$. While $z \notin [-1,1]$ and when $N \to \infty$, $\rho(z)^{-N}$ and $q_N(z) \to 0$, so

$$q_0(z) = \frac{f(z)}{\sqrt{\mu_0}} = -\sum_0^\infty \frac{\varepsilon_n(z)}{\rho(z)^{n+1}} = \frac{2\sqrt{\mu_0}}{\rho(z)} + \sum_0^\infty \frac{(2a_n-1)(q_{n-1}(z) + \rho(z)q_n(z)) + 2b_nq_n(z)}{\rho(z)^{n+1}}$$

From the known asymptotic behaviour $q_n(z) \sim \dfrac{2(2\pi)^{1/2}D(\rho(z)^{-1})}{(\rho(z)-\rho(z)^{-1})\rho(z)^n}$ for large n [Ba], we have a first approximate expansion of $(2\pi)^{-1/2}D(\rho(z)^{-1})(\rho(z)-\rho(z)^{-1})q_0(z)$ as a series of negative powers of $\rho(z)$:

$$(2\pi)^{-1/2}D(\rho(z)^{-1})(\rho(z)-\rho(z)^{-1})q_0(z) \approx \sum_0^\infty \frac{2(2a_n-1)}{\rho(z)^{2n}} + \frac{2b_n}{\rho(z)^{2n+1}}$$

Under sufficiently strong conditions , this remains valid when $z = x+i\varepsilon, \varepsilon \to 0, \varepsilon > 0, -1 \le x = \cos(\theta) \le 1$, $\rho(z) \to e^{i\theta}$ (with $0 \le \theta \le \pi$), so that $2a_n - 1$ and b_n behave like the *Fourier coefficients* of order $2n$ and $2n+1$ of a function whose singularities on $0 < \theta < \pi$ are related to the singularities of $w(\cos\theta)\sin\theta$. Singularities of type $|\theta-\theta_0|^\gamma$ correspond indeed to $n^{-1-\gamma}\cos(n\theta_0 +$ const.) behaviour in the n^{th} Fourier coefficient.

This analysis is no more valid for stronger singularity $w(x_0) = 0$ or ∞ because the approximations done near x_0 are no more valid, subtle important effects take place in neighbourhoods of length about $1/n$ of the singular points, see the famous [NevIII].

It is therefore not useless to have a close look at the simplest orthogonal polynomials related to weights with interior singularities. Special singular positions have been worked (in sieved polynomials theory etc.), but here is something related to an *arbitrary* position:

2.Freud's equations for the simplest generalized Jacobi polynomials.

Let

$$\begin{aligned} w(x) &= B(1-x)^{\alpha}(x_0-x)^{\gamma}(1+x)^{\beta}\ x\in[-1,x_0]\ , \\ &= A(1-x)^{\alpha}(x-x_0)^{\gamma}(1+x)^{\beta}\ x\in[x_0,1]\ , \end{aligned} \tag{3}$$

with $-1<x_0<1$, A and $B>0$, α, β and $\gamma>-1$.

The case $A=B, \alpha=\beta, x_0=0$ has the simple solution deduced from Jacobi polynomials $a_n^2 = (n+2\alpha+\gamma\,\mathrm{odd}\,(n))(n+\gamma\,\mathrm{odd}\,(n))/[(2n+2\alpha+\gamma+1)(2n+2\alpha+\gamma-1)], b_n = 0$, where $\mathrm{odd}\,(n) = (1-(-1)^n)/2$. When $n\to\infty$, this case shows the asymptotic behaviour $a_n \sim \dfrac{1}{2} - \dfrac{(-1)^n\gamma}{4n}, b_n = 0$.

The $O(1/n)$ term is definitely related to the $|x|^{\gamma}$ behaviour of the weight near 0, as shown by P. Nevai (Theorem 4 of Section 7 of [Nev]): if w is even on $[-1,1]$, with $w(x)|x|^{-\gamma}$ positive and continuously derivable on $(-1,1)$, then $a_n = 1/2-\gamma(-1)^n/(4n)+o(1/n)$ when $n\to\infty$.

We now try to investigate the recurrence coefficients a_n, b_n when the weight is (3). This weight is a *semi-classical* weight, as w'/w is the same rational function almost everywhere on the support $[-1,1]$ of w.

Semi-classical orthogonal polynomials have a rich differential structure, according to a theory going as far as Laguerre [BeR, GaN, Lag, Mag2, Mag3, Sho].

Freud [Fr] showed how to deduce recurrence coefficients asymptotics from special identities. For a general semi-classical weight satisfying $w'(x)/w(x) = 2V(x)/W(x)$ with $W(x)w(x)\to 0$ when x tends to any endpoint of the support S of w (Shohat's conditions [Sho]), we find these identities (*Freud's equations*) by expanding

$$\begin{aligned} 0 &= \int_S [W(x)w(x)p_n(x)p_{n-k}(x)]'\,dx \\ &= \int_S W(x)w(x)p_n'(x)p_{n-k}(x)\,dx + \int_S W(x)w(x)p_n(x)p_{n-k}'(x)\,dx + \\ &\qquad + \int_S W'(x)w(x)p_n(x)p_{n-k}(x)\,dx + \int_S 2V(x)w(x)p_n(x)p_{n-k}(x)\,dx \end{aligned}$$

for $k=0,1$, remarking that any integral $\int_S P(x)w(x)p_n(x)p_{n-k}(x)dx$ where P is a polynomial, is an expression involving $a_n, b_n, a_{n\pm1}, b_{n\pm1}$, etc. according to k and the degree of P, and using $p_n' = np_{n-1}/a_n + (b_0+\cdots+b_{n-1}-nb_{n-1})p_{n-2}/(a_{n-1}a_n)+\cdots$ (see [BeR]).

Reduction to even measure: one considers the orthonormal polynomials $\{\tilde p_n\}$ with respect to the even weight

$$\tilde w(x) = 2|x|w(2x^2-1), \qquad \text{for } -1<x<1.$$

Then, $\tilde p_{2n}(x) = p_n(2x^2-1)$, and one recovers the recurrence relation for the p_n's by contracting the recurrence relation for the $\tilde p_n$'s:

$$\begin{aligned} &\tilde a_{n+1}\tilde p_{n+1}(x) = x\tilde p_n(x) - \tilde a_n\tilde p_{n-1}(x) \qquad \Rightarrow \\ &\qquad \tilde a_{2n+1}\tilde a_{2n+2}\tilde p_{2n+2}(x) = (x^2-\tilde a_{2n}^2-\tilde a_{2n+1}^2)\tilde p_{2n}(x) - \tilde a_{2n-1}\tilde a_{2n}\tilde p_{2n-2}(x)\ , \end{aligned}$$

so:

$$a_n = 2\tilde{a}_{2n-1}\tilde{a}_{2n}\ , \qquad b_n = -1 + 2\tilde{a}_{2n}^2 + 2\tilde{a}_{2n+1}^2 \tag{4}$$

This allows to work with the single sequence $\{\tilde{a}_n\}$ instead of the two sequences $\{a_n\}, \{b_n\}$.

Here,

$$\begin{aligned}\tilde{w}(x) = 2|x|w(2x^2-1) &= \tilde{B}|x|^{2\beta+1}(\tilde{x}_0^2 - x^2)^\gamma(1-x^2)^\alpha \qquad \text{for } |x| < |\tilde{x}_0|,\\ &= \tilde{A}|x|^{2\beta+1}(x^2 - \tilde{x}_0^2)^\gamma(1-x^2)^\alpha \qquad \text{for } |\tilde{x}_0| < |x| < 1,\end{aligned}$$

where $\tilde{x}_0$ is the positive root of $2\tilde{x}_0^2 - 1 = x_0$, $\tilde{A} = 2^{\alpha+\beta+\gamma+1}A$, $\tilde{B} = 2^{\alpha+\beta+\gamma+1}B$.

So, $\tilde{W}(x) = x(x^2 - \tilde{x}_0^2)(x^2-1) = x^5 - (\tilde{x}_0^2+1)x^3 + \tilde{x}_0^2 x$ and $2\tilde{V}(x) = (2\alpha+2\beta+2\gamma+1)x^4 - [2\alpha\tilde{x}_0^2 + (2\beta+1)(\tilde{x}_0^2+1) + 2\gamma]x^2 + (2\beta+1)\tilde{x}_0^2$.

The equations for the $\tilde{a}_n$'s now follow from Freud's method for even weights, expanding $\tilde{W}\tilde{w}' = 2\tilde{V}\tilde{w}$ as

$$\int_S \frac{\tilde{W}(x)}{x}\tilde{w}(x)\tilde{p}_n'(x)\tilde{p}_{n-1}(x) + \int_S \frac{\tilde{W}(x)}{x}\tilde{w}(x)\tilde{p}_n(x)\tilde{p}_{n-1}'(x) + \int_S \left(\frac{\tilde{W}(x)}{x}\right)'\tilde{w}(x)\tilde{p}_n(x)\tilde{p}_{n-1}(x) + \\ + \int_S \frac{2\tilde{V}(x)}{x}\tilde{w}(x)\tilde{p}_n(x)\tilde{p}_{n-1}(x) = 0,$$

using $\int_S \tilde{w}(x)\tilde{p}_i(x)\tilde{p}_j(x)dx = \delta_{i,j}$, the recurrence relations (1) giving $\int_S x\tilde{w}(x)\tilde{p}_n(x)\tilde{p}_{n-1}(x)dx = \tilde{a}_n$, $\int_S x^2\tilde{w}(x)(\tilde{p}_n(x))^2dx = \tilde{a}_n^2 + \tilde{a}_{n+1}^2$, $\int_S \tilde{w}(x)x^{-1}\tilde{p}_n(x)\tilde{p}_{n-1}(x)dx = \text{ odd}\,(n)/\tilde{a}_n$, etc., and
$\tilde{p}_n' = \dfrac{n}{\tilde{a}_n}\tilde{p}_{n-1} + \dfrac{2\sum_1^{n-1}\tilde{a}_k^2 - n\tilde{a}_{n-1}^2}{\tilde{a}_{n-2}\tilde{a}_{n-1}\tilde{a}_n}\tilde{p}_{n-3} +$
$+\dfrac{n\tilde{a}_{n-3}^2\tilde{a}_{n-1}^2 - 2(\tilde{a}_{n-3}^2 + \tilde{a}_{n-2}^2 + \tilde{a}_{n-1}^2)\sum_1^{n-1}\tilde{a}_k^2 + 2\sum_1^{n-1}(\tilde{a}_k^4 + 2\tilde{a}_k^2\tilde{a}_{k-1}^2)}{\tilde{a}_{n-4}\tilde{a}_{n-3}\tilde{a}_{n-2}\tilde{a}_{n-1}\tilde{a}_n}\tilde{p}_{n-5} + \cdots\ ,$
one finally finds

$$2(n+\alpha+\beta+\gamma+2)\tilde{a}_n^2(\tilde{a}_{n-1}^2 + \tilde{a}_n^2 + \tilde{a}_{n+1}^2) - 2[\alpha\tilde{x}_0^2 + (n+\beta+1)(\tilde{x}_0^2+1) + \gamma]\tilde{a}_n^2 + \\ + 2(2\tilde{a}_n^2 - \tilde{x}_0^2 - 1)\sum_{j=1}^{n-1}\tilde{a}_j^2 + n\tilde{x}_0^2 - 2\tilde{a}_n^2\tilde{a}_{n-1}^2 + 2\sum_{j=1}^{n-1}(\tilde{a}_j^4 + 2\tilde{a}_j^2\tilde{a}_{j-1}^2) + (2\beta+1)\tilde{x}_0^2\,\text{odd}\,(n) = 0, \\ n = 1, 2, \ldots \tag{5}$$

$(\tilde{a}_0 = 0)$.

We see how any $\tilde{a}_n$ can be computed from the value of $\tilde{a}_1$, which is the degree of freedom reflecting that the same equations (5) hold for any choice of $\tilde{A}$ and $\tilde{B}$ in the weight $\tilde{w}$. Actually $\tilde{a}_1$ is linked to the ratio $\tilde{A}/\tilde{B}$ by

$$\tilde{a}_1^2 = \frac{\tilde{\mu}_2}{\tilde{\mu}_0} = \frac{\tilde{B}\int_{|x|<\tilde{x}_0}|x|^{2\beta+3}(\tilde{x}_0^2 - x^2)^\gamma(1-x^2)^\alpha\,dx + \tilde{A}\int_{|x|>\tilde{x}_0}|x|^{2\beta+3}(x^2-\tilde{x}_0^2)^\gamma(1-x^2)^\alpha\,dx}{\tilde{B}\int_{|x|<\tilde{x}_0}|x|^{2\beta+1}(\tilde{x}_0^2 - x^2)^\gamma(1-x^2)^\alpha\,dx + \tilde{A}\int_{|x|>\tilde{x}_0}|x|^{2\beta+1}(x^2-\tilde{x}_0^2)^\gamma(1-x^2)^\alpha\,dx} \tag{6}$$

Numerical experiments show that $\tilde a_2$, $\tilde a_3$,... can be computed in a stable way from $\tilde a_1$ simply by considering (5) as an equation for $\tilde a_{n+1}$ when $\tilde a_1$, ..., $\tilde a_n$ are known:

$$\tilde a_{n+1}^2 = \frac{\alpha\tilde x_0^2 + (n+\beta+1)(\tilde x_0^2+1)+\gamma}{N} - 2\frac{\sum_1^{n-1}\tilde a_k^2}{N} + \\ + \frac{2(\tilde x_0^2+1)\sum_1^{n-1}\tilde a_k^2 - n\tilde x_0^2 - 2\sum_1^{n-1}(\tilde a_k^4 + 2\tilde a_k^2\tilde a_{k-1}^2) - (2\beta+1)\tilde x_0^2 \,\mathrm{odd}\,(n)}{2N\tilde a_n^2} + \frac{\tilde a_{n-1}^2}{N} - \tilde a_n^2 - \tilde a_{n-1}^2 \tag{7}$$

with $N = n+\alpha+\beta+\gamma+2$, $n = 1, 2, \ldots$.

3.Asymptotic estimates.

Some people call these tricks "special refinements";
others call them "kludges".
D.E. Knuth

Putting an almost constant $\tilde a_n^2 \approx \tilde a^2$ in (5) gives two possible asymptotic matches $\tilde a^2 = 1/4$ and $\tilde a^2 = \tilde x_0^2/4$, corresponding to weights with support $[-1, 1]$ and $[-\tilde x_0, \tilde x_0]$ (the latter when $A = 0$). More complicated behaviours are expected to hold when $\tilde x_0$ is complex [GaN] and one should be able to establish correct asymptotic behaviours from (5) alone, but this has not yet been achieved (the answer to [GaN] was given in [N] with other techniques). Even when one knows that $\tilde a_n^2 \to 1/4$, some amount of guesswork will still be needed. Let $\tilde a_n^2 = \dfrac{1}{4} + y_n$. We know from Szegő's theory ([Sz] chap. 12) that $\sum_1^{n-1}\tilde a_k^2 = \dfrac{n}{4} + \xi + z_n$ and $\sum_1^{n-1}(\tilde a_k^4 + 2\tilde a_k^2\tilde a_{k-1}^2) = \dfrac{3n}{16} + \eta + u_n$ with z_n and $u_n \to 0$ (See also [Nev2] p.91). So,

$$y_{n+1} = -\frac{1}{4} + \frac{\alpha\tilde x_0^2 + (n+\beta+1)(\tilde x_0^2+1)+\gamma}{N} - 2\frac{\dfrac{n}{4}+\xi+z_n}{N} + \\ + \frac{2(\tilde x_0^2+1)(\dfrac{n}{4}+\xi+z_n) - n\tilde x_0^2 - 2(\dfrac{3n}{16}+\eta+u_n) - (2\beta+1)\tilde x_0^2\,\mathrm{odd}\,(n)}{2N(\dfrac{1}{4}+y_n)} + \frac{\dfrac{1}{4}+y_{n-1}}{N} - \frac{1}{2} - y_n - y_{n-1}.$$

We compute now ξ and η: from the Szegő theory, let $\tilde\phi_n(z) = \tilde\kappa_n z^n + \tilde\kappa'_n z^{n-2} + \tilde\kappa''_n z^{n-4} + \cdots$ be the orthonormal polynomials on the unit circle with respect to $|\sin\theta|\tilde w(cos\theta) = \tilde C(\theta)|\cos\theta|^{2\beta+1}|\cos^2\theta - \cos^2(\theta_0/2)|^\gamma|\sin\theta|^{2\alpha+1}$, with $\tilde C(\theta) = \tilde A$ on $|\theta| < \theta_0/2$ and $|\theta-\pi| < \theta_0/2$ and $\tilde C(\theta) = \tilde B$ elsewhere. The Szegő function $\tilde D(z)$ whose boundary values must be $|\tilde D(e^{i\theta})| = \sqrt{\tilde w(\cos\theta)|\sin\theta|}$ is found by inspection to be

$$\tilde D(z) = \tilde\kappa^{-1}(1-z^2)^{\alpha+1/2}(1+z^2)^{\beta+1/2}(1-e^{-i\theta_0}z^2)^{\gamma/2+i\lambda}(1-e^{i\theta_0}z^2)^{\gamma/2-i\lambda},$$

with $\tilde\kappa = 2^{\alpha+\beta+\gamma+1}\tilde B^{(\theta_0-\pi)/(2\pi)}\tilde A^{-\theta_0/(2\pi)}$ and $\lambda = (2\pi)^{-1}\log(\tilde B/\tilde A)$. We find the limit values of $\tilde\kappa'_n$ and $\tilde\kappa''_n$ from the expansion of $1/\tilde D$:

$$\tilde\kappa'_n/\tilde\kappa_n \to \alpha - \beta + \gamma x_0 + 2\lambda\sin\theta_0,$$

$$\tilde\kappa''_n/\tilde\kappa_n \to (\alpha-\beta)^2/2 + (\alpha+\beta+1)/2 + (\alpha-\beta)(\gamma x_0 + 2\lambda\sin\theta_0) + \\ + (\gamma(\gamma+2)/4 - \lambda^2)\cos(2\theta_0) + \lambda(\gamma+1)\sin(2\theta_0) + \gamma^2/4 + \lambda^2,$$

used in

$$\widetilde{P}_n(x) = \frac{\tilde{p}_n(x)}{\tilde{\gamma}_n} =$$
$$= x^n - \left(\sum_1^{n-1} a_k^2\right) x^{n-2} + \left[\left(\sum_{k=1}^{n-1} \tilde{a}_k^2\right)^2 - \left(\sum_{k=1}^{n-1} \tilde{a}_k^4\right) - 2\left(\sum_{k=1}^{n-2} \tilde{a}_k^2 \tilde{a}_{k+1}^2\right)\right] x^{n-4}/2 + \cdots$$
$$= \frac{z^{-n}\tilde{\phi}_{2n}(z) + z^n \tilde{\phi}_{2n}(z^{-1})}{2^n(\tilde{\kappa}_{2n} + \tilde{\phi}_{2n}(0)}$$
$$\sim \frac{T_n(x)}{2^{n-1}} + \frac{\tilde{\kappa}'}{\tilde{\kappa}}\frac{T_{n-2}(x)}{2^{n-1}} + \frac{\tilde{\kappa}''}{\tilde{\kappa}}\frac{T_{n-4}(x)}{2^{n-1}} + \cdots$$
$$\sim x^n - \frac{n - \tilde{\kappa}'/\tilde{\kappa}}{4} x^{n-2} + \frac{n(n-3)/2 - (n-2)\tilde{\kappa}'/\kappa + \tilde{\kappa}''/\tilde{\kappa}}{16} x^{n-4} + \cdots$$

$(z + z^{-1} = 2x)$.

So,

$$\xi = \lim_{n\to\infty} \sum_1^{n-1} \tilde{a}_k^2 - n/4 = -\kappa'/(4\kappa) = -(\alpha - \beta + \gamma x_0 + 2\lambda \sin\theta_0)/4, \tag{8}$$

$\eta = \lim_{n\to\infty} \sum_1^{n-1}(\tilde{a}_k^4 + 2\tilde{a}_k^2\tilde{a}_{k-1}^2) - 3n/16 = ((\kappa'/\kappa)^2 - 4\kappa'/\kappa - 2\kappa''/\kappa)/16$.

The equation for the y_n's reduces to

$$\begin{aligned} y_{n+1} - 2x_0 y_n + y_{n-1} = {} & \frac{(x_0+1)(\beta+1/2)(-1)^n + 2(x_0+2)z_n + y_{n-1} - 4u_n}{N} - \\ & - \frac{16 y_n}{1+4y_n} \frac{(2x_0+1)/4 - \lambda\sin\theta_0 + (x_0+1)(\beta+1/2)(-1)^n + 2(x_0+3)z_n - 4u_n}{4N} \\ & - \frac{4(2x_0+1)y_n^2}{1+4y_n} \end{aligned} \tag{9}$$

where $z_n = -\sum_n^\infty y_k$ and $u_n = -y_{n-1}/2 - \sum_n^\infty (3y_k/2 + y_k^2 + 2y_k y_{k-1})$.

The form (9) should give hints on the behaviour of y_n when $n \to \infty$. No proof will be attempted here, only reasonable asymptotic matching and numerical checks. Use of Painlevé-like differential equations in θ_0 is another method of investigation which could be used in the future (see [Mag3]).

The right-hand side is small, even with respect to the y's, so $y_{n+1} - 2x_0 y_n + y_{n-1}$ is small, and this suggests a $\exp(\pm in\theta_0)$ behaviour somewhere. However, I still don't have a tight proof that y_n, z_n and u_n are $O(1/n)$. Assuming $y_n = K_1(-1)^n/n + K_2 e^{in\theta_0}/n^{\zeta_2} + K_3 e^{-in\theta_0}/n^{\zeta_3}$, matching the two sides gives $K_1 = -\beta/2 - 1/4$, $\zeta_{2,3} = 1 \pm 2i\lambda$. Numerical checks have been performed on the form

$$\tilde{a}_n^2 - 1/4 = y_n = -(\beta + 1/2)(-1)^n/(2n) + K\cos(n\theta_0 - 2\lambda\log n - \varphi)/n + o(1/n) \tag{10}$$

where K and φ are unknown functions of α, β, γ, λ and x_0. Given α, β, γ and x_0, the algorithm first performs (7) with several trial starting values $\tilde{a}_1^2$ and computes the corresponding λ from (8), allowing the determination of the coefficients in (6). It is then possible to run (7) for a requested value of λ, and to estimate K and φ in (10) from numerical values of $\tilde{a}_n^2$ for large n (up to the 10000-100000 range). Very satisfactory empirical formulas for K and φ appear to be $K=(\gamma^2/4+\lambda^2)^{1/2}\sin(\theta_0/2)$ and $\varphi=(\alpha+1+\gamma/2)\pi-(\alpha+\beta+\gamma+1/2)\theta_0+2\lambda\log(2\sin\theta_0)-2\arg\Gamma(\gamma/2+i\lambda)-\arg(\gamma/2+i\lambda)$.

Whence, from (4): $a_n-1/2\sim y_{2n-1}+y_{2n}, b_n\sim 2(y_{2n}+y_{2n+1})$, the

Conjecture. The recurrence coefficients related to the simplest generalized Jacobi weight (3) satisfy

$$a_n=\frac{1}{2}-\frac{M}{n}\cos\left[2n\theta_0-2\lambda\log(4n\sin\theta_0)-\Phi\right]+o\left(\frac{1}{n}\right),$$
$$b_n=-\frac{2M}{n}\cos\left[(2n+1)\theta_0-2\lambda\log(4n\sin\theta_0)-\Phi\right]+o\left(\frac{1}{n}\right),$$

when $n\to\infty$, where $x_0=\cos\theta_0$, $0<\theta_0<\pi$, $\lambda=\log(B/A)/(2\pi)$, $M=\frac{1}{2}(\gamma^2/4+\lambda^2)^{1/2}\sin\theta_0$, $\Phi=(\alpha+\gamma/2)\pi-(\alpha+\beta+\gamma)\theta_0-2\arg\Gamma(\gamma/2+i\lambda)-\arg(\gamma/2+i\lambda)$.

Here is a sample of the numerical check: K and φ are extracted from the form (10) on a sample of $\tilde{a}_n^2$, with n going up to 500000. The values of K and λ (from (8)) are quite stable, but things are not so easy with φ (the "phase" column). Finally, the line "check" contains the computed values of K and φ, and a new check of λ through Turán determinant weight function reconstruction yielding $\tilde{A}$ and $\tilde{B}$.

Acknowledgements.

Many thanks to H.Dette, L.Golinskii and P.Nevai for their interest. The computations have been performed on the Convex `C3820` of the University.

Many thanks to G. Allasia who manages to edit the present proceedings.

References

- Ap A.I. APTEKAREV, Asymptotics of orthogonal polynomials in a neighborhood of the endpoints of the interval of orthogonality, *Russ. Acad. Nauk Mat. Sb.* **183** (1992) = *Russian Acad. Sci. Sb. Math.* **76** (1993) 35-50.
- Ba W. BARRETT, An asymptotic formula relating to orthogonal polynomials, *J. London Math. Soc. (2)* **6** (1973), 701-704.
- BeR S. BELMEHDI, A. RONVEAUX, About non linear systems satisfied by the recurrence coefficients of semiclassical orthogonal polynomials, to appear in *J. Approx. Th.*
- Fr G.FREUD, On the coefficients in the recursion formulæ of orthogonal polynomials, *Proc. Royal Irish Acad. Sect. A* **76** (1976), 1-6.
- GaN J.L. GAMMEL, J. NUTTALL, Note on generalized Jacobi polynomials, *in* "The Riemann Problem, Complete Integrability and Arithmetic Applications" (D. Chudnovsky and G. Chudnovski, Eds.), pp.258-270, Springer-Verlag (Lecture Notes Math. **925**), Berlin, 1982.
- Ho C.H. HODGES, Van Hove singularities and continued fraction coefficients, *J. Physique Lett.* **38** (1977), L187-L189.

- Lag E. LAGUERRE, Sur la réduction en fractions continues d'une fraction qui satisfait à une équation différentielle linéaire du premier ordre dont les coefficients sont rationnels, *J. Math. Pures Appl. (4)* **1** (1885), 135-165 = pp. 685-711 *in Oeuvres*, Vol.II, Chelsea, New-York 1972.
- LaG Ph. LAMBIN, J.P. GASPARD, Continued-fraction technique for tight-binding systems. A generalized-moments approach, *Phys. Rev. B* **26** (1982) 4356-4368.
- Mag1 A.P. MAGNUS, Recurrence coefficients for orthogonal polynomials on connected and non connected sets, *in* "Padé Approximation and its Applications, Proceedings, Antwerp 1979" (L. Wuytack, editor), pp. 150-171, Springer-Verlag (Lecture Notes Math. **765**), Berlin, 1979.
- Mag2 A.P. MAGNUS, On Freud's equations for exponential weights, *J. Approx. Th.* **46** (1986) 65-99.
- Mag3 A.P. MAGNUS, Painlevé-type differential equations for the recurrence coefficients of semi-classical orthogonal polynomials, to appear in *J. Comp. Appl. Math.*
- Mart A.MARTIN *et al.*, Léon Van Hove, 1924-1990, *Courrier CERN* **30** n°2 (March 1991), 20-27.
- Nev P.NEVAI, *Orthogonal Polynomials, Memoirs AMS* vol. **18** nr. **213** (March 1979).
- NeIII P.NEVAI, Mean convergence of Lagrange interpolation III, *Trans. Amer. Math. Soc.* **282** (1984) 669-698.
- Nev2 P. NEVAI, Orthogonal polynomials, recurrences, Jacobi matrices and measures, pp. 79-104 *in Progress in Approximation Theory, an International Perspective* (A.A.GONCHAR & E.B.SAFF, editors), Springer-Verlag, New York, 1992.
- NV P.NEVAI, W. VAN ASSCHE, Compact perturbations of orthogonal polynomials, *Pacific J. Math.* **153** (1992) 163-184.
- N J.NUTTALL, Asymptotics of generalized Jacobi polynomials, *Constr. Approx.* **2** (1986) 59-77.
- Sho J.A. SHOHAT, A differential equation for orthogonal polynomials, *Duke Math. J.* **5** (1939),401-417.
- Sz G. SZEGŐ, *Orthogonal Polynomials*, Colloquium Publications, Vol. 23 Amer. Math. Soc. Providence, Rhode Island, 1967.
- VA W. VAN ASSCHE, Lecture Notes, Namur, Feb. 1993.


```
Script started on Wed Mar  2 17:38:49 1994
/u18/grpanma/magnus/jacobi@ux12 [1] #ls -l
total 1224

-rwxr-xr-x  1 magnus     473298 Mar  2 17:30 a.out
-rw-r--r--  1 magnus       2226 Feb 16 15:24 alnrel.f
-rwxr-xr-x  1 magnus         18 Jan  3 14:15 batchsub
-rw-r-----  1 magnus       1183 Feb 16 08:58 c8lgmc.fz
-rw-r-----  1 magnus       2087 Feb 16 08:58 c9lgmc.f
-rw-r--r--  1 magnus        259 Feb 16 15:25 carg.f
-rw-r-----  1 magnus      21751 Feb 16 16:41 clngam.f
-rw-r--r--  1 magnus      14412 Feb 16 16:42 clngam.o
-rw-r-----  1 magnus       2645 Feb 16 15:39 clngamold.f
-rw-r-----  1 magnus        975 Feb 16 08:58 clnrel.f
```

```
-rw-r--r--  1 magnus       141 Feb 16 16:35 cmplxt.f
-rw-r--r--  1 magnus      1061 Feb 16 15:52 csevl.f
-rw-r-----  1 magnus       471 Feb 16 08:58 entsrc.fz
-rw-r--r--  1 magnus      7679 Mar  2 17:29 gjacobi.f
-rw-r--r--  1 magnus     14554 Mar  2 17:30 gjacobi.o
-rw-r--r--  1 magnus      6394 Feb  9 09:07 gjacobi4.f
-rw-r-----  1 magnus      4153 Feb 16 08:59 indexfn.z
-rw-r--r--  1 magnus       861 Feb 16 15:52 inits.f
-rw-r--r--  1 magnus       167 Feb 16 16:46 lgam.f
-rw-r--r--  1 magnus      1325 Feb 16 16:46 lgam.o
-rwxr-xr-x  1 magnus    452258 Mar  2 13:50 lgam.out
-rw-r-----  1 magnus     10873 Feb 16 08:58 macharb.f
-rw-r-----  1 magnus      4759 Feb 16 08:58 r1macha.fz
-rw-r-----  1 magnus      8318 Feb 16 08:58 r1machb.f
-rw-r-----  1 magnus      2800 Feb 16 08:58 seterr.f
-rw-r-----  1 magnus       319 Feb 16 08:58 seteru.f

/u18/grpanma/magnus/jacobi@ux12 [2] #cat gjacobi.f
c  gjacobi.for
      program gjacobi
c simplest generalized Jacobi
c
c    w(x) =  B (1-x)^alpha  (1+x)^beta  (x0-x)^gamma   -1<x<x0
c         =  A (1-x)^alpha  (1+x)^beta  (x-x0)^gamma   x0<x<1
c
c   ->  w tilde(y)= B tilde |y|^(2beta+1) (y0-y)^gamma (1-y^2)^alpha
c                                                            -y0<|y|<y0
c   ->  w tilde(y)= A tilde |y|^(2beta+1) (y-y0)^gamma (1-y^2)^alpha
c                                                             y0<|y|<1
c
c x=2y^2-1
c
c the program produces squares of recurrence coefficients atilde n
c
c   of   atilde n+1 ptilde n+1 (y) = y ptilde n (y) - atilde n ptilde n-1 (y)
c
c   where the ptilde s are the orthonormal polynomials related to w tilde
c
c input: first, period of display of atilde n, number of values of y
c            where to compute polynomials, and interval (this latter part
c            is alpha test, enter 0 0 0)
c
c    then: alpha, beta, gamma, y0 , n max
c
c    then, after preliminary calculations have been done,
c              lambda = log(B/A) /(2 pi)
```

```
c
      integer oddn,n,nmax,i,signg,recomp,period,npol
      double precision al,be,ga,y0,atn,atnp1,pi,aux,shalf,cth,sth,
     &           at2,at2p1,at2m1,one,half,
     &           th0, x, fn, cnt,snt,s1,s2,snp1,cnp1,xx,xy,yy,lam1,lam2
     &           ,aux1,at,bt,ba(4),at21(4),c1,c2,c3,thx0,lam0,phi0,phi
     &           ,phi1,lsth
c   or, if you can afford it  (useful if |lambda|>5) :
c     real*16 al,be,ga,y0,atn,atnp1,pi,aux,shalf,cth,sth,
c    &           at2,at2p1,at2m1,one,half,
c    &           th0, x, fn, cnt,snt,s1,s2,snp1,cnp1,xx,xy,yy,lam1,lam2
c    &           ,aux1,at,bt,ba(4),at21(4),c1,c2,c3,thx0,lam0,phi0,phi
c    &           ,phi1,lsth
      double precision xv(502),polv(502,3),dtur
c   complex gamma function from W. Fullerton's fnlib
c        available by anonymous ftp on netlib.att.com    cd/netlib/fn
      complex clngam,z
      one=1
      half=one/2
      pi=4*atan(one)
      print *,'  period of display of coeff., number of abs., extr. abs.? '
      read *,period,npol,xv1,xv2
c  recurrence coefficients will be printed with step period,
c  polynomials and weight will be computed at xv1,...,xv2
      npol=min(npol,500)
1     print *,' input alpha,beta,gamma,xtilde0,nmax  (stop if nmax<=0)'
      read *,al,be,ga,y0,nmax
      if(nmax.le.0)stop
      print '(4f14.8,i8,f9.4)',al,be,ga,y0,nmax
      signg=1
      if(ga.lt.0.0)signg=-1
      shalf=sqrt(1-y0*y0)
      cth=2*y0*y0-1
      sth=2*shalf*y0
      th0=acos(cth)
      phi0=pi*(2*al+2+ga)/2  - (al+be+ga+half)*th0
      lsth=log(2*sth)
      xv(1)=y0/2
      xv(2)=(1+y0)/2
      do 100 i=1,npol
  100 xv(i+2)=xv1+(i-1)*(xv2-xv1)/(npol-0.999)
      itba=1
      at2=y0*y0
      at21(1)=at2
c 3 satisfactory trial values of atilde square(1) are needed to
c establish the homographic relation atilde square(1)=c1+c2B/(A+c3B)
```

```
 11   if(itba.eq.1)goto 12
      if(itba.le.3)then
       if(at21(itba).le.0)at21(itba)=0.0001
       if(at21(itba).ge.1)at21(itba)=0.9999
       at21(itba)=0.99*at21(itba)+0.01*y0*y0
c to be tried until all the atilde square(n) are positive
       at2=at21(itba)
      else
      print *,' ? lambda  ( stop if > 999)'
       read *, lam0
       if(lam0.gt.999)goto 1
       ba(4)=exp(2*pi*lam0)
       at2=c1+c2/(c3+1/ba(4))
       at21(4)=at2
      endif
   12 cnt=cth
      snt=sth
      s1=0
      s2=0
      at2m1=0
      oddn=1
      if(itba.ge.4)
     &print *,'   n     atilde2(n)       K       phase    lambda  lambda'
      if(itba.ge.4)print '(i8,f14.10)',1,at2
      atn=sqrt(at2)
      do 13 i=1,npol+2
      polv(i,1)=1
   13 polv(i,2)=xv(i)/atn
c  main engine for recurrence coefficients
      do 10 n=1,nmax
       aux=al*y0*y0+(n+be+1)*(y0*y0+1)+ga+( (y0*y0+1)/at2 -2 )*s1
     &     -(n*y0*y0)/(2*at2)+at2m1-s2/at2 -(be+half)*y0*y0*oddn/at2
       at2p1=aux/(n+al+be+ga+2) -at2m1-at2
c updating the two sums
       s1=s1+at2
       s2=s2+at2*( at2+2*at2m1 )
       snp1=snt*cth+cnt*sth
       cnp1=cnt*cth-snt*sth
c check asymptotic formula when needed to display
       if((n.lt.2).or.(at2p1.lt.0).or.(mod(n+1,period).eq.0).or.
     &       (n.eq.nmax)) then
        aux=(n+1)*(at2p1-half/2)+(oddn-half)*(be+half)
c aux should be about K cos( (n+1)theta0 -phi )
c with phi slowly varying with n
        aux1=n*(at2-half/2)-(oddn-half)*(be+half)
c  K cos(phi):
```

```
        xx=(aux1*snp1-aux*snt)/sth
c  K sin(phi):
        yy=(aux*cnt-aux1*cnp1)/sth
c phi  (between -pi and pi )
        phi=atan2(yy,xx)
c K
        xx=sqrt(xx*xx+yy*yy)
c  2 ways to estimate lambda
        lam1=(-4*s1+n+1-al+be-ga*cth)/(2*sth)
        lam2=half*(-16*s2+3*(n+1)-5*al+3*be-1-ga*(2*cth*cth+4*cth-1))/
     &       (2*sth*(cth+2))
       fn=n+1
c  phi -2 lambda log n
       fn=phi-2*lam1*log(fn)
       fn=fn/(2*pi)
       ifn=fn
       fn=fn-ifn
       if(fn.lt.-half)fn=fn+1
       if(fn.gt. half)fn=fn-1
       phi=2*pi*fn
c  Theta n(x tilde 0):
c     thx0=4*s1+2*(n+al+be+ga)*at2+2*(n+2+al+be+ga)*at2p1
c    &     -2*(n+be+ga+1-ga*y0*y0)
c     thx0=thx0*y0
      if(itba.ge.4)
     &print '(i8,f14.10,4f10.5)',n+1,at2p1,xx,phi,lam1,lam2
       endif
       if(at2p1.le.0)then
           if(itba.ge.4)
     &      print *,' atildesquare <0, try a less extreme lambda'
            goto 11
       endif
c values of orthogonal polynomial of degree n+1
       atnp1=sqrt(at2p1)
       do 25 i=1,npol+2
   25  polv(i,3)=( xv(i)*polv(i,2)-atn*polv(i,1) )/atnp1
c prepare next step
      if(n.lt.nmax)then
       snt=snp1
       cnt=cnp1
       oddn=1-oddn
       at2m1=at2
       at2=at2p1
       atn=atnp1
       do 251 i=1,npol+2
       polv(i,1)=polv(i,2)
```

```
 251  polv(i,2)=polv(i,3)
     endif
 10  continue
     if(itba.ge.4)then
c check : weight function reconstruction
c w(x)= 2 sqrt(1-x^2) /(pi * dtur)
     do 20 i=1,npol+2
       x=xv(i)
      dtur=polv(i,2)**2-atnp1*polv(i,1)*polv(i,3)/atn
      fn=1/(dtur*pi* x**(2*be+1) *(abs(y0*y0-x*x))**ga
    &       *(1-x*x)**(al-half))
      if(i.eq.1)bt=fn
      if(i.eq.2)at=fn
      if(i.gt.2)print '(1p,5e15.7)',x,fn,(polv(i,j),j=1,3)
  20 continue
c     print '('' Atilde='',1p,e12.4, '' Btilde='',e12.4,''  B/A ='',
c    &  e12.4)',at,bt,bt/at
      phi1=phi0+2*lam1*lsth
      z=cmplx(ga/2,lam1)
      z=clngam(z)
      phi1=phi1-2*aimag(z)-atan(2*lam1/ga)
      phi1=phi1+(1-signg)*pi/2
      phi1=phi1/(2*pi)
     iphi1=phi1
     phi1=phi1-iphi1
     if(phi1.gt.half)phi1=phi1-1
     if(phi1.lt.-half)phi1=phi1+1
     phi1=phi1*2*pi
     print '('' check:'',15x,3f10.5,1p,2e10.3)',
    &           shalf*sqrt(ga**2+4*lam1**2)*half,
    &           phi1,half*log(bt/at)/pi,at,bt
     endif
c estimate of B/A = exp(2pi lambda)
      ba(min(itba,4))=exp(2*pi*lam1)
      if(itba.lt.4)print '(i3,f15.8)',itba,at21(itba)
      itba=itba+1
      recomp=0
      if(itba.eq.4)recomp=1
c  recompute the homographic relation if new extreme data have been found
      if(itba.gt.4)then
        if(at21(4).lt.at21(2))then
          recomp=1
          at21(2)=at21(4)
          ba(2)=ba(4)
        endif
        if(at21(4).gt.at21(3))then
```

```
          recomp=1
          at21(3)=at21(4)
          ba(3)=ba(4)
        endif
      endif
      if(recomp.eq.1)then
       xx=(at21(1)-at21(2))/(1/ba(1)-1/ba(2))
       yy=(at21(1)-at21(3))/(1/ba(1)-1/ba(3))
       c3=(1/ba(3)-1/ba(2))/(xx/yy-1)-1/ba(2)
       c2=(at21(1)-at21(2))*(1/ba(1)+c3)*(1/ba(2)+c3)/(1/ba(2)-1/ba(1))
       c1=at21(1)-c2/(1/ba(1)+c3)
       print '(''('',f15.6,'' A+'',f15.6,'' B)/(A+ '',f15.6,'' B)'')',
     &                    c1,c1*c3+c2,c3
       print '('' a1 tilde square min,max= '',2f10.5)',c1,c1+c2/c3
      endif
      if(itba.eq.2)at21(2)=0
      if(itba.eq.3)at21(3)=1
      goto 11
      end
/u18/grpanma/magnus/jacobi@ux12 [3] #a.out

   period of display of coeff., number of abs., extr. abs.?
50000 0 0 0
  input alpha,beta,gamma,xtilde0,nmax  (stop if nmax<=0)
0.1 -0.66 -0.9 0.8 500000
    0.10000000   -0.66000000   -0.90000000    0.80000000  500000
  1     0.64000000      2     0.48642906    3     0.66612028
(       0.666359 A+       0.682666 B)/(A+        1.404457 B)
 a1 tilde square min,max=    0.66636   0.48607
  ? lambda  ( stop if > 999)     1
    n     atilde2(n)       K       phase    lambda  lambda
       1  0.4863105269
       2  0.1072928956  0.56893   0.63299  -0.23606  -0.11591
   50000  0.2500104918  0.65794   2.04846   1.00002   1.00003
  100000  0.2499993374  0.65795   2.04921   0.99999   1.00000
  150000  0.2500017530  0.65794   2.04881   1.00001   1.00001
  200000  0.2499986110  0.65795   2.04911   1.00000   1.00000
  250000  0.2500025306  0.65795   2.04887   1.00001   1.00001
  300000  0.2499980788  0.65795   2.04903   1.00000   1.00000
  350000  0.2500018867  0.65796   2.04897   1.00000   1.00001
  400000  0.2499994423  0.65795   2.04895   1.00000   1.00001
  450000  0.2499999531  0.65795   2.04904   1.00000   1.00000
  500000  0.2500011150  0.65795   2.04892   1.00001   1.00001
  500001  0.2500009765  0.65795   2.04898   1.00000   1.00001
 check:                 0.65795   2.04898   1.00000 6.315E-05 3.382E-02
(       0.666359 A+       0.682666 B)/(A+        1.404457 B)
```

```
 a1 tilde square min,max=    0.66636   0.48607
  ? lambda  ( stop if > 999)     -2
    n     atilde2(n)       K       phase    lambda  lambda
       1  0.6663578695
       2  0.0057517038   0.88852   1.15619  -0.61116  -0.49518
   50000  0.2500181130   1.22984  -2.15364  -1.99944  -1.99940
  100000  0.2500124101   1.22976  -2.15333  -1.99945  -1.99943
  150000  0.2500050485   1.22973  -2.15338  -1.99945  -1.99944
  200000  0.2499972423   1.22969  -2.15318  -1.99946  -1.99946
  250000  0.2499956982   1.22970  -2.15294  -1.99947  -1.99947
  300000  0.2500015654   1.22970  -2.15304  -1.99947  -1.99946
  350000  0.2500034533   1.22971  -2.15326  -1.99946  -1.99945
  400000  0.2499985243   1.22969  -2.15320  -1.99946  -1.99946
  450000  0.2499982891   1.22970  -2.15304  -1.99947  -1.99946
  500000  0.2500023390   1.22969  -2.15318  -1.99946  -1.99946
  500001  0.2499993553   1.22970  -2.15306  -1.99947  -1.99946
 check:                  1.22969  -2.15322  -1.99946 4.756E-02 1.664E-07
(       0.666359 A+       0.682666 B)/(A+        1.404457 B)
 a1 tilde square min,max=    0.66636   0.48607
  ? lambda  ( stop if > 999)      3
    n     atilde2(n)       K       phase    lambda  lambda
       1  0.4860711314
       2  0.1073891907   0.56857   0.63221  -0.23556  -0.11549
   50000  0.2499658427   1.90240   1.71797   3.13855   3.13861
  100000  0.2499817770   1.90236   1.71765   3.13858   3.13861
  150000  0.2499992879   1.90235   1.71679   3.13862   3.13864
  200000  0.2500085823   1.90243   1.71702   3.13861   3.13864
  250000  0.2499927104   1.90240   1.71719   3.13861   3.13862
  300000  0.2500063733   1.90242   1.71677   3.13862   3.13864
  350000  0.2499951650   1.90239   1.71715   3.13861   3.13862
  400000  0.2500046810   1.90242   1.71677   3.13862   3.13864
  450000  0.2499960453   1.90241   1.71706   3.13861   3.13862
  500000  0.2500039648   1.90243   1.71682   3.13862   3.13863
  500001  0.2500009249   1.90243   1.71704   3.13862   3.13863
 check:                  1.90243   1.71683   3.13864 9.229E-11 3.386E-02
(       0.666359 A+       0.682666 B)/(A+        1.404457 B)
 a1 tilde square min,max=    0.66636   0.48607
  ? lambda  ( stop if > 999)     -4
    n     atilde2(n)       K       phase    lambda  lambda
       1  0.6663587554
       2  0.0057510998   0.88853   1.15619  -0.61116  -0.49518
   50000  0.2500178596   2.17836  -0.90688  -3.60233  -3.60219
  100000  0.2499851215   2.17817  -0.90842  -3.60225  -3.60221
  150000  0.2499982856   2.17817  -0.90898  -3.60222  -3.60219
  200000  0.2499966056   2.17815  -0.90894  -3.60222  -3.60220
  250000  0.2499928901   2.17814  -0.90874  -3.60223  -3.60221
```

```
  300000  0.2499936079   2.17816  -0.90855  -3.60224  -3.60222
  350000  0.2500002389   2.17818  -0.90867  -3.60223  -3.60221
  400000  0.2500054650   2.17815  -0.90898  -3.60222  -3.60220
  450000  0.2500020824   2.17816  -0.90913  -3.60221  -3.60220
  500000  0.2499962590   2.17814  -0.90891  -3.60222  -3.60221
  500001  0.2500006090   2.17815  -0.90911  -3.60221  -3.60220
 check:                  2.17813  -0.90913  -3.60221 4.756E-02 7.042E-12
(       0.666359 A+       0.682666 B)/(A+        1.404457 B)
 a1 tilde square min,max=    0.66636   0.48607
  ? lambda  ( stop if > 999)     5
    n     atilde2(n)       K       phase    lambda  lambda
       1  0.4860711310
       2  0.1073891909   0.56857   0.63221  -0.23556  -0.11549
      26 -0.1229105534   9.79980  -0.13175   4.66254   4.89361
  atildesquare <0, try a less extreme lambda
  ? lambda  ( stop if > 999)     4.5
    n     atilde2(n)       K       phase    lambda  lambda
       1  0.4860711310
       2  0.1073891909   0.56857   0.63221  -0.23556  -0.11549
   50000  0.2500552560   2.75176  -0.28737   4.56431   4.56451
  100000  0.2499780806   2.75188  -0.28663   4.56429   4.56436
  150000  0.2499892472   2.75186  -0.28715   4.56432   4.56437
  200000  0.2500094861   2.75176  -0.28772   4.56435   4.56439
  250000  0.2500050796   2.75188  -0.28704   4.56432   4.56437
  300000  0.2499915823   2.75187  -0.28722   4.56433   4.56435
  350000  0.2500079170   2.75187  -0.28764   4.56435   4.56438
  400000  0.2499943758   2.75181  -0.28714   4.56433   4.56435
  450000  0.2500045768   2.75184  -0.28771   4.56435   4.56437
  500000  0.2499965741   2.75183  -0.28722   4.56433   4.56435
  500001  0.2499948277   2.75188  -0.28741   4.56434   4.56436
 check:                  2.75188  -0.28769   4.56436 1.188E-14 3.386E-02
(       0.666359 A+       0.682666 B)/(A+        1.404457 B)
 a1 tilde square min,max=    0.66636   0.48607
  ? lambda  ( stop if > 999)     5
    n     atilde2(n)       K       phase    lambda  lambda
       1  0.4860711310
       2  0.1073891909   0.56857   0.63221  -0.23556  -0.11549
   50000  0.2499526977   3.01064  -1.08241   4.99743   4.99760
  100000  0.2500207551   3.01066  -1.08330   4.99749   4.99762
  150000  0.2500144656   3.01067  -1.08379   4.99752   4.99761
  200000  0.2499949927   3.01057  -1.08351   4.99751   4.99757
  250000  0.2499902230   3.01067  -1.08422   4.99754   4.99758
  300000  0.2500095958   3.01063  -1.08471   4.99757   4.99760
  350000  0.2499959458   3.01060  -1.08402   4.99754   4.99757
  400000  0.2500006139   3.01058  -1.08470   4.99757   4.99759
  450000  0.2500018568   3.01067  -1.08429   4.99755   4.99758
```

```
  500000  0.2499976049   3.01065  -1.08462   4.99757   4.99758
  500001  0.2500043587   3.01063  -1.08476   4.99757   4.99759
 check:                  3.01067  -1.08480   4.99756 7.809E-16 3.386E-02
(       0.666359 A+       0.682666 B)/(A+        1.404457 B)
 a1 tilde square min,max=    0.66636   0.48607
  ? lambda  ( stop if > 999)     9999
  input alpha,beta,gamma,xtilde0,nmax  (stop if nmax<=0)
0 0 0 0 0 0
STOP:
/u18/grpanma/magnus/jacobi@ux12 [4] #exit
exit

script done on Wed Mar  2 18:01:22 1994
```